Explications
V2 premier lifting :
Renommage des feuilles avec des noms courts et non accentués.
Dans la feuilles BD_Pers : Nommage des colonnes A et D (Ligne 1)
Dans la feuilles BD_Date : Création d'une plage Feries avec des dates verticales
[bookmark: _GoBack]Cette plage est encore incomplète et ne sera finalisé que dans la V6.xlsb

	On pourrait prévoir de réunir ces deux feuilles (BD_Pers et BD_Date) ultérieurement dans une feuille unique (Param ?)

Dans le gestionnaire de noms :
Suppression des noms accentués 
Modification des Noms existants avec des formules dynamiques (avec DECALER)
Création de nouveaux noms :
BDW Fait référence à : =DECALER(BD!$A$1;;;NBVAL(BD!$A:$A);4)
NameW Fait référence à : =DECALER(BD!$A$1;1;;NBVAL(BD!$A:$A)-1)
DateW Fait référence à : =DECALER(BD!$B$1;1;;NBVAL(BD!$B:$B)-1)
iDW Fait référence à : =DECALER(BD!$C$1;1;;NBVAL(BD!$C:$C)-1)
DoW Fait référence à : =DECALER(BD!$D$1;1;;NBVAL(BD!$D:$D)-1)
Pers Fait référence à : =DECALER(BD_Personnel!$A$1;1;;NBVAL(BD_Personnel!$A:$A)-1)
Feries Fait référence à : =DECALER(BD_Date!$A$1;1;;NBVAL(BD_Date!$A:$A)-1)
Cette dernière est inutilisée pour l'instant et ne sera finalisé que dans la V6.xlsb

	+ diverses modifications mineures sur des formules existantes qui seront supprimées dans les versions suivantes.

Dans la feuille Recap :
Utilisation des nouveaux Noms créés :
On travaille sur la Ligne 12 et quand toutes les formules sont Ok on fait un Copier/Collage Spécial Formules vers le bas.
En C12 : 
	=SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)*(DoW=6)*(iDW=C$10))
En D12 :
	=SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)*(DoW=6)*(iDW=D$10))
En E12 :
	=C12+D12
En G12 :
	=SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)*(DoW=7)*(iDW=G$10))
En H12 :
	=SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)*(DoW=7)*(iDW=H$10))
En I12 :
	=SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)*(DoW=7)*(iDW=I$10))
En J12 :
	=G12+H12+I12
En L12 :
	=SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)*((DateW=Fériés_N)+(DateW=Fériés_N1))*((iDW=1)+(iDW=2)))
En O12 :
	=N12-SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)*(iDW="C"))
en R12 :
	=Q12-SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(DateW<=FIN)*(NameW=$A12)*(iDW="RTT")))) *Cette formule est fausse ce qui est sans conséquence mais sera corrigé ultérieurement
Correction : =Q12-SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN) NameW=$A12)*(iDW="RTT"))))
En T12 :
=SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(DateW<=FIN)*(NameW=$A12)*(iDW=T$10)) *Cette formule est fausse ce qui est sans conséquence mais sera corrigé ultérieurement
Correction : =SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)*(iDW=T$10))
En U12 :
=SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(DateW<=FIN)*(NameW=$A12)*(iDW=U$10)) *Cette formule est fausse ce qui est sans conséquence mais sera corrigé ultérieurement
Correction : =SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)*(iDW=U$10))
En V12 :
=SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(DateW<=FIN)*(NameW=$A12)*(iDW=V$10)) *Cette formule est fausse ce qui est sans conséquence mais sera corrigé ultérieurement
Correction : =SOMMEPROD((DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)*(iDW=V$10))

A ce stade quand toutes les formules de la ligne 12 sont Ok on fait un Copier/Collage Spécial Formules vers le bas.

	On remarque que toutes les formules comportent une même valeur logique qui sera exploitée par la suite :
(DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)

Dans VBA :
Premier renommage des variables avec des noms courts et non accentués.
Typage des variables : Toutes les variables qui se terminent par % = as Integer
Première optimisation en utilisant la définition de Pers pour dimensionner la variable (iLR) et les pointeurs (i) et (iRow)
Ajout des données sur la 4ème colonne avec Weekday(...)
Le code ne sera quasiment plus modifié par la suite sauf dans la V6.xlsb ou on travaillera sur un Array au lieu de travailler directement sur la feuille BD ce qui permet de supprimer le recalcul manuel et accélère le traitement.

Dans la V4 et V6 (il y peu de différences) :
Je suis repassé avec une version en ligne de mes Feries de manière à valider l'idée d'une formule simple avant de trouver une meilleure solution.
Ne pas trop en tenir compte ce point est solutionné dans la V6.xlsb

L'important dans ces versions est surtout de bien comprendre la modification majeure :
On a remarqué plus haut que pour la ligne 12 toutes les formules comportent une même valeur logique : 
(DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)
Nous allons donc "encapsuler" cette valeur dans un Nom Défini (FRMDC) du gestionnaire de nom.
C'est très simple on place le curseur sur C12 et on définit un nouveau nom (j'ai choisi "FRMDC" par habitude mais "libellule" ou "galopin" ferait aussi bien l'affaire...)
Et dans la zone fait référence à on écrit :
=(DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)
YAPUKA modifier toutes les cellules de la ligne 12 pour intégrer la potion magique ; ce qui évite au proc de calculer à chaque fois (DateW>=DEBUT)*(DateW<=FIN)*(NameW=$A12)

ça donne donc :
En C12 : 
=SOMMEPROD(FRMDC*(DoW=6)*(iDW=C$10))
En D12 :
=SOMMEPROD(FRMDC*(DoW=6)*(iDW=D$10))
En E12 :
=C12+D12
En G12 :
=SOMMEPROD(FRMDC*(DoW=7)*(iDW=G$10))
En H12 :
=SOMMEPROD(FRMDC*(DoW=7)*(iDW=H$10))
En I12 :
=SOMMEPROD(FRMDC*(DoW=7)*(iDW=I$10))
En J12 :
=G12+H12+I12
En L12 :
=SOMMEPROD(FRMDC*(DateW=Feries)*((iDW=1)+(iDW=2)))
En O12 :
=N12-SOMMEPROD(FRMDC*(iDW="C"))
en R12 :
=Q12-SOMMEPROD(FRMDC*(iDW="RTT"))
En T12 :
=SOMMEPROD(FRMDC(iDW=T$10))
En U12 :
=SOMMEPROD(FRMDC(iDW=U$10))
En V12 :
=SOMMEPROD(FRMDC(iDW=V$10))
A ce stade quand toutes les formules de la ligne 12 sont Ok on fait un Copier/Collage Spécial Formules vers le bas.

Dans la V6.xlsb
J'ai trouvé une solution à mon problème de Feries (et corrigé le calendrier pour le rendre permanent quel que soit l'année)
A noter dans cette feuille (BD_Date) que j'ai également remplacé le Nom année par "An" (Il faut bannir les accents dans les Noms, Noms de feuille et le VBA)
Ce qui modifie la feuille Recap de la manière suivante :
En L12 :
	=SOMMEPROD(FRMDC*(NB.SI(Feries;DateW)) *((iDW=1)+(iDW=2)))
(ne pas oublier le Copier/Collage Spécial Formules vers le bas)

Enfin dans VBA j'ai remplacé le travail dans la feuille "BD" par un travail sur Array (C'est la variable Arr)
Une toute petite modif dans VBA mais qui accélère le traitement et permet de supprimer le recalcul manuel puisqu'on ne travaille plus sur cette feuille (on injecte seulement les données de la variable Arr en bloc en fin de W)

Elle est pas belle la vie ?

